Second Order Linear Equations and the Airy Functions:

Why Special Functions are Really No More Complicated than Most Elementary Functions

We shall consider here the most important second order ordinary differential equations, namely linear equations. The standard format for such an equation is

$$y''(t) + p(t)\, y'(t) + q(t)\, y(t) = g(t),$$

where y(t) is the unknown function satisfying the equation and p, q and g are given functions, all continuous on some specified interval. We say that the equation is homogeneous if g = 0. Thus:

$$y''(t) + p(t)\, y'(t) + q(t)\, y(t) = 0.$$

We have studied methods for solving an inhomogeneous equation for which you have already solved the corresponding homogeneous equation. So we shall concentrate on homogeneous equations here. The equation is said to be a constant coefficient equation if the functions p and q are constant. Again we have studied methods for dealing with those. Non-constant coefficient equations are more problematic, but alas, they arise frequently in nature (e.g., in simple mechanical and electrical systems). In this lesson we shall study closely one of the best known examples -- Airy's Equation. For the record, the solutions to that equation, i.e., the Airy functions, arise in diffraction problems in the study of optics, and also in relation to the famous Schroedinger equation in quantum mechanics.

Before proceeding, let's recall some basic facts about the set of solutions to a linear, homogeneous second order differential equation. The most basic fact is that the set of solutions forms a two-dimensional vector space. This means that you can find two solutions, y_1 and y_2, neither of which is a multiple of the other, so that all solutions are given by linear combinations of these two:

$a\, y_1 + b\, y_2$, where a and b are arbitrary constants.

Just to help you get your bearings, let's mention two other two-dimensional vector spaces that you know very well:

The Euclidean plane, where every vector can be expressed as

$a\,\mathbf{i} + b\,\mathbf{j}$

where $\mathbf{i} = (1,0)$ and $\mathbf{j} = (0,1)$;

or all polynomials of degree at most 1

$a + b\,x$.

Now returning to second order linear homogenous differential equations with constant coefficients, we note, by way of examples, that all solutions of

$$y'' + y = 0$$

are given by

a cos(t) + b sin(t);

and all solutions of

y'' +(1/t)y' + (1/t^2)y = 0

are given by

a cos(ln(t)) + b sin(ln(t)).

(The latter is an Euler equation.) In the first example, the interval of
definition is the whole real line, whereas in the second, we must restrict to t > 0. In both instances, a and b are arbitrary real numbers.

Airy's Equation

This is the equation:

y" - ty = 0.

In this presentation we shall solve it symbolically and numerically. We shall also address it graphically. Furthermore, if you look at DEwM, Problem Set D, Problem 1, p. 145, you will see a qualitative method of dealing with the equation, which I shall briefly recall below. Finally, although it is not in the course syllabus, one can also use the method of series solution to solve Airy's Equation. So there are lots of ways to skin this cat.

Well, let's try **DSolve** first and see what happens.

```
DSolve[y''[t] - t * y[t] == 0, y[t], t]
```

```
{{y[t] → AiryAi[t] C[1] + AiryBi[t] C[2]}}
```

How about that, *Mathematica* solves it. But what are those functions it reports as the answers? In fact, they are *special functions*, and if I may quote from DEwM, p. 46:

"By special functions we mean various non-elementary functions that mathematicians give names to, often because they arise as solutions of particularly important differential equations." Recall also that elementary functions are "the standard functions of calculus: polynomials, exponentials and logarithms, trigonometric functions and their inverses, and all combinations of these functions through algebraic operations and compositions." The simplest special function is the *error function*:

$$\mathrm{erf}(t) = (2/\sqrt{\pi})\int_0^t e^{-s^2} ds.$$

Many special functions have integral formulas like the above and/or are specified by a differential equation and/or are given by a power series.

Graphing Airy Functions

```
? AiryAi
? AiryBi
```

AiryAi[z] gives the Airy function Ai(z). ≫

AiryBi[z] gives the Airy function Bi(z). ≫

These are two linearly independent solutions of Airy's equation. We can learn more about them by clicking on the "more info" double carets above.

OK, let's graph these functions.

```
Plot[AiryAi[t], {t, -2, 2}]
```



Not so helpful; let's plot on a bigger interval.

```
Plot[AiryAi[t], {t, -10, 10}]
```



Looks like a mildly damped oscillation to the left, but with increasing frequency; and a rapid decay to zero to the right. Let's examine the second solution.

```
Plot[AiryBi[t], {t, -2, 2}]
```



and again on a larger interval that reveals more about the function:

```
Plot[AiryBi[t], {t, -10, 3}]
```



So both solutions manifest decreasing oscillations toward minus infinity, apparently with increasing frequencies, but as $t \to +\infty$, one solution grows without bound and the other decays to zero.

- **Some Qualitative Analysis**

Now I recall Problem 1 in PSD in DEwM. It leads the reader through the following reasoning. For a large negative value of t, say t = - K^2, Airy's Equation resembles

y" + K^2 y = 0,

whose solutions are sinusoidal oscillations with frequency K. Thus the oscillatory behavior on the negative axis of the Airy functions is not surprising. Moreover, as t moves toward -∞, and so has larger absolute value (K is getting bigger), then the frequency is increasing. This analysis does not allow us to conclude anything about the amplitude.

On the other hand, for t large postive, say near K^2, the equation resembles

y" - K^2 y = 0,

whose solutions are e^(Kt) and e^(-Kt). Thus the growth at positive infinity is expected; and exactly as the functions

$a\, e^t + b\, e^{-t}$

have exactly "one direction" in which the function decays at +∞, whereas in all other directions the solutions grow quickly, it is again not surprising that the same behavior is manifested by the Airy functions.

Indeed, the basic Airy function AiryAi(t) = Ai(t) is exactly that special choice among the Airy functions.

- **Numerical solutions to yield a graphical presentation**

Now we imitate the code on p. 132 of DEwM. As we saw above, there are two arbitrary constants to be specified in the choice of an Airy function. That corresponds to the fact that the second order Airy equation requires two pieces of initial data to determine a specific solution. Thus drawing a representative set of solutions does not, like in the case of first order equations, yield a set of non-intersecting curves. Still, sometimes the pictures are striking and reveal the general nature of solutions rather dramatically -- see e.g., the graph on p. 132 of DEwM.

```
ode = y''[t] - t * y[t] == 0;
nsol := NDSolve[{ode, y[0] == y0, y'[0] == yp0}, y[t], {t, -2, 2}];
nsolfunc[t_, y0_, yp0_] := y[t] /. First[nsol]
Plot[Evaluate[Table[nsolfunc[t, y0, yp0], {y0, -2, 2}, {yp0, -1, 1, 0.5}]],
 {t, -2, 2}, PlotRange → {-10, 10}, AxesOrigin → {-2, 0}]
```



Really nice. Let’s expand the sols both to the left and right to see if we can see the damped oscillation or increased frequencies to the left and/or the exp growth to the right.

```
ode = y''[t] - t * y[t] == 0;
nsol := NDSolve[{ode, y[0] == y0, y'[0] == yp0}, y[t], {t, -7, 2}];
nsolfunc[t_, y0_, yp0_] := y[t] /. First[nsol]
Plot[Evaluate[Table[nsolfunc[t, y0, yp0], {y0, -2, 2}, {yp0, -1, 1, 0.5}]],
 {t, -7, 2}, PlotRange → {-10, 10}, AxesOrigin → {-2, 0}]
```



```
ode = y''[t] - t * y[t] == 0;
nsol := NDSolve[{ode, y[0] == y0, y'[0] == yp0}, y[t], {t, -2, 5}];
nsolfunc[t_, y0_, yp0_] := y[t] /. First[nsol]
Plot[Evaluate[Table[nsolfunc[t, y0, yp0], {y0, -2, 2}, {yp0, -1, 1, 0.5}]],
 {t, -2, 5}, PlotRange → {-20, 20}, AxesOrigin → {-2, 0}]
```



Note that in all of these pictures, we picked up the zero solution, but not the solution Ai(t). That is not surprising because

```
Ai0 = AiryAi[0]
Aip0 = AiryAi'[0]
```

$1 \Big/ \left(3^{2/3}\,\text{Gamma}\left[\frac{2}{3}\right]\right)$

$-1 \Big/ \left(3^{1/3}\,\text{Gamma}\left[\frac{1}{3}\right]\right)$

Those were not among the initial values that we selected to generate our plots. (Note I named these because I will use them below.)

An aside on the Gamma Function if you are not familiar with it. By definition, the Gamma Function is defined by the integral formula: $\Gamma(z) = \int_0^\infty t^{z-1} e^{-t}\, dt$. It has many interesting features, but perhaps the best known is that Γ(n) = (n -1)!

■ Stability

We have not considered stability of second order equations, but it is not hard to envision what we would mean -- small perturbations in the initial data -- both position and velocity -- should lead to only small perturbations in the solution curve over the long term. Given what we have learned about the Airy functions, do you think the Airy equation is stable?

Not likely; just the form of the equation y'' = ty and the first order derivative test suggests not. Let's see if we can justify that assertion. We know that Ai(t) is the only solution of Airy's equation that decays at infinity. Let's solve the equation numerically and compare what we get (graphically) to the curve of Ai(t):

```
SymAiryPlot = Plot[AiryAi[t], {t, 0, 10}, PlotRange → {0, 0.4}]
```



I have named the plot, which will allow me to recall this figure later.

```
nsol2 := NDSolve[{ode, y[0] == Ai0, y'[0] == Aip0}, y[t], {t, 0, 10}];
nsolfunc2[t_] := y[t] /. First[nsol2]
NumAirPlot = Plot[Evaluate[nsolfunc2[t]], {t, 0, 10}, PlotRange → {0, 10}]
```



Can you explain the graph?

The Nature of Special Functions

Finally, I would like to convince you that special functions like the Airy Function are really no more mysterious than many of our elementary functions -- like the sine, exponential or logarithm.

In fact, how do you define sin(x)? In most caclulus books, the function sin(x) is defined by saying: let x measure an angle in radians, then draw a right triangle with that angle, whereupon sin(x) is the ratio of the length of the opposite side over the hypotenuse. If I ask you to tell me what sin(sqrt(3)) is, do you really have a good feeling for that value? So, many advanced calculus books attempt to put the definition of the sine function on a firmer analytic footing. They define the sine function as follows. First define the function ArcSin(t) by the integral formula:

$$\int_0^t \frac{1}{\sqrt{1-s^2}}\, ds, \ -1 < t < 1.$$

Then they engage in some calculus to establish that this is a differentiable function, monotone increasing, with

ArcSin(-1) = -π/2, ArcSin(1) = π/2,
and a vertical tangent line at the two endpoints. Finally, they define sin(x) to be the inverse function of ArcSin(t), which is then defined on [-π/2, π/2] and finally they extend it to the whole real line by invoking periodicity.

Now is that any simpler than the method we have used to define Ai(t)? I won't go through the derivation but I will tell you that the basic Airy function can also be obtained by an integral. Here is the formula:

$$\text{Ai}(t) = (1/\pi) \int_0^\infty \cos\left((1/3)\, s^3 + \, ts\right) ds.$$

Not an easy integral, but we can deal with it numerically when necessary. This process is perhaps a little more complicated technically than:

$$\ln(t) = \int_0^t 1/s\; ds,$$
exp(t) = inverse function of ln(t);

but aren't we talking about more or less the same kind of object.

The moral of the story: we can deal with Ai(t) and most special functions in the same way that we deal with elementary functions:
graphically, numerically, analytically and even symbolically on occasion.
You should not be intimidated by special functions.

There are more special functions than you can imagine:
Bessel functions, Legendre functions, hypergeometric functions, the Riemann Zeta function and a score more. Many of these, but not all, are solutions of second order homogeneous non-constant coefficient
equations. You may encounter some of them in physics or engineering courses. Let's just see if *Mathematica* can deal with the Bessel functions. Those are the solutions of the equation:

t^2 y''(t) + t y'(t) + (t^2 - n^2) y(t) = 0

for different choices of an integer n.

```
DSolve[t^2 * y''[t] + t * y'[t] + (t^2 - n^2) * y[t] == 0, y[t], t]
```

```
{{y[t] → BesselJ[n, t] C[1] + BesselY[n, t] C[2]}}
```

I leave it to you to explore other special functions in *Mathematica*, but let us just draw the fundamental solutions of the Bessel equation for n = 0.

```
Plot[BesselJ[0, t], {t, -6, 6}]
```



Looks like a sinusoidal; let's redraw on a bigger interval.

```
Plot[BesselJ[0, t], {t, -10, 10}]
```



and even bigger interval

```
Plot[BesselJ[0, t], {t, -100, 100}]
```



Damped oscillation clearly. Actually, the oscillatory behavior is not surprising. For t large, the middle term in the equation t y’’ + y’ + t y =0

is negligible and so the sinusoidal behavior is evident. Also, it looks like the solution curve has the same behavior in both directions. In fact bessel functions of order zero are even. That’s a good exercise: show that if y(t) satisfies the Bessel equation (with n = 0), then so
does y(-t).

Now for the other solution:

```
Plot[BesselY[0, t], {t, -6, 6}]
```



That' s interesting; why no negative values? In fact, *Mathematica* declines to present any values of the Bessel function for negative t because BesselY[0, t] has a logarithmic singularity at t = 0. So *Mathematica* discards the negative values, which are just a mirror reflection of the positive values. Let' s see the singularity more clearly.

```
Plot[BesselY[0, t], {t, 0, 0.01}]
Plot[BesselY[0, t], {t, 0, 0.00001}]
```





In fact lim BesselY[0, t] as t $\longrightarrow 0^+$ is - ∞; although the divergence is very slow (logarithmic).

OK, I can’t resist, let' s look at one more example -- the hypergeometric equation:

```
DSolve[t * (1 - t) * y''[t] + (c - (1 + a + b) * t) * y'[t] - a * b * y[t] == 0, y[t], t]
```

```
{{y[t] → C[1] Hypergeometric2F1[a, b, c, t] +
    (-1)^(1-c) t^(1-c) C[2] Hypergeometric2F1[1 + a - c, 1 + b - c, 2 - c, t]}}
```

solves the hypergeometric equation. I leave further experimentation with special functions as solutions of second order homogeneous equations to you; there are lots of examples in Boyce & DiPrima. In fact, most of the examples appear in the chapter on series solutions. Since that topic is not covered in our syllabus, I will have to leave it to you to look at on your own or to encounter in other math, physics or engineering courses.